

# Parametric surfaces

or, the cross-product and surface integrals

On the [Wikipedia page for parametric surfaces](#), we see the lovely formula

$$\int_S 1dS = \int \int_D \|\mathbf{r}_u \times \mathbf{r}_v\| dudv.$$

This states that the area of a parametrized surface can be computed as the double integral of the magnitude of a normal vector to the surface. It forms a special case of a *surface integral*, which is a central object of study for us, and forms yet another reason to discuss the cross product! After exploring this formula, we'll be in a great position to explore more general surface integrals.

## Surface parametrizations

### The basic idea

From one perspective, a surface parametrization is simply one way to visualize a function mapping  $\mathbb{R}^2 \rightarrow \mathbb{R}^3$ . The basic idea is to start with a given subset  $D$  of the domain space, parametrized by input variables say  $u$  and  $v$ , and visualize it's image  $T(D)$  in the range space, parametrized by output variables  $x$ ,  $y$ , and  $z$ . This technique is illustrated in figure 1.

We emphasize that figure 1 is *not* a graph of the function  $T$  in the technical sense. Figure 1 is an illustration of the image under  $T$  of one subset of the plane. This is analogous to visualizing a function  $f : \mathbb{R} \rightarrow \mathbb{R}$  using the effect of  $f$  on a sub-divided interval in the line. This is illustrated in figure 2, where we see the image of 20 evenly distributed points throughout the interval  $[0, 2]$  under the function  $f(x) = x^2$ . One point to notice in figure 2 is that the points are no longer evenly distributed after the function is applied - that is, application of the function can distort lengths.

### Graphs of functions over various domains

A parametric surface could be considered to be a generalization of the idea of the graph of a function. Specifically, the graph of a function  $f : \mathbb{R}^2 \rightarrow \mathbb{R}^3$ , is precisely the set of all  $(x, y, z) \in \mathbb{R}^3$  such that  $z = f(x, y)$ . Thus, we might define  $T(u, v) = (u, v, f(u, v))$  or even, since the correspondence between  $(u, v)$  and  $(x, y)$  is so direct, as  $T(x, y) = (x, y, f(x, y))$ . This basic idea is illustrated in figure 3.

A major application of these types transformations, however, is that this allows us to plot graphs over other types of regions. For example, suppose we want to plot the function




Figure 1: Mapping the unit square to a surface in  $\mathbb{R}^3$ .




Figure 2: The image of a collection of points under  $f(x) = x^2$




Figure 3: The plot of  $f(x, y) = e^{-4(x^2+y^2)} + 1$  over a square

$$f(x, y) = e^{-4(x^2+y^2)} + 1$$

over the unit disk. We could do so by parametrically plotting the function

$$T(r, \theta) = (r \cos(\theta), r \sin(\theta), e^{-4r^2} + 1).$$

This is illustrated in figure 4. Do you see why this works?




Figure 4: A plot of  $f(x, y) = e^{-4(x^2+y^2)} + 1$  over the unit disk

## Tubes

Even more generally, we are not restricted to considering surfaces generated by graphs of functions. Consider, for example,

$$T(u, v) = \langle u, \cos(v), \sin(v) \rangle = \langle 1, 0, 0 \rangle u + \langle 0, 1, 0 \rangle \cos(v) + \langle 0, 0, 1 \rangle \sin(v).$$

The little rewrite allows us to see what's going on. The first component describes motion along the  $x$ -axis. The combined effect yields a tube around the  $x$ -axis, as shown in figure 5.

This idea can be extended. We can parametrize a tube around any curve, as long as we know three vectors as a function of where we are on the curve:




Figure 5: The parametrization of a tube

- $\mathbf{T}$  a unit vector in the direction of motion,
- $\mathbf{N}$  a unit vector that's perpendicular to the direction of motion, and
- $\mathbf{B}$  a unit vector that's perpendicular to both  $T$  and  $N$ .

For example, a torus can be parametrized by

$$p(s, t) = \langle \cos(t), \sin(t), 0 \rangle + \frac{1}{2} \langle \cos(t), \sin(t), 0 \rangle \cos(s) + \frac{1}{2} \langle 0, 0, 1 \rangle \sin(s),$$

as shown in figure 6.

### The TNB frame

Given a parametrization  $\mathbf{r}(t)$ , there is a general procedure for finding vectors  $\mathbf{T}$ ,  $\mathbf{N}$ , and  $\mathbf{B}$  that work. We simply let

- $\mathbf{T}(t) = \mathbf{r}'(t) / \|\mathbf{r}'(t)\|$ ,
- $\mathbf{N}(t) = \mathbf{T}'(t) / \|\mathbf{T}'(t)\|$ , and
- $\mathbf{B}(t) = \mathbf{T}(t) \times \mathbf{N}(t)$ .

We'll put this to work, once we start playing with code!

### Spherical coordinates

Another class of parametrizations arises from spherical coordinates, which it's time to start talking about! The spherical coordinate transformations are




Figure 6: The parametrization of a torus.

$$\begin{aligned}x &= \rho \sin(\varphi) \cos(\theta) \\y &= \rho \sin(\varphi) \sin(\theta) \\z &= \rho \cos(\varphi)\end{aligned}$$

The spherical coordinates transformations relate  $x$ ,  $y$ , and  $z$  to *three* input variables, namely  $\rho$ ,  $\varphi$ , and  $\theta$ . We can generate a parametric surface by setting one of these to a constant, or even to a function of the other two. For example, if we set  $\rho = 1$ , we obtain a parametrization of a sphere - an application of tremendous importance in geography! If we set  $\rho$  to a function of  $\varphi$  and  $\theta$ , we might generate a lumpy sphere. Figure 7, for example, shows a portion of the lumpy sphere obtained when  $\rho = (\sin(5u) \cos(5v) + 4)/4$ . I wonder what the area of that is?

## Generating parametric plots on the computer

Our textbook has a Sage cell that shows you how to generate 3D parametric plots with Sage. *Mathematica* can also generate 3D parametric plots with the fabulous `ParametricPlot3D` command. The syntax looks like so:

```
ParametricPlot3D[{x[u, v], y[u, v], z[u, v]}, {u, a, b}, {v, c, d}]
```

For example, we can plot the groovy part of figure 4 like so:

```
ParametricPlot3D[{r * Cos[t], r * Sin[t], Exp[-4 r^2] + 1},
  {r, 0, 1}, {t, 0, 2 Pi}]
```

Here's the code for a sphere:




Figure 7: A portion of a lumpy sphere.

```
ParametricPlot3D[
  {Cos[u] Sin[v], Sin[u] Sin[v], Cos[v]},
  {u, 0, 2 Pi}, {v, 0, Pi}]
```

Or our lumpy sphere:


```
ParametricPlot3D[
  (Sin[5 u] Cos[5 v] + 4) {Cos[u] Sin[v], Sin[u] Sin[v], Cos[v]}/4,
  {u, -Pi/2, Pi/2}, {v, Pi/12, Pi/2}]
```

And a torus!

```
ParametricPlot3D[
  {Cos[t], Sin[t], 0} + 0.5 {Cos[t], Sin[t], 0} Cos[s] + 0.5 {0, 0, 1} Sin[s],
  {t, 0, 2 Pi}, {s, 0, 2 Pi}]
```

You can have Mathematica do the work of computing a TNB frame for you to generate a tube around almost any curve. Here's a fabulous example called a Trefoil knot.

```
trefoil = {Sin[3 t], Sin[t] + 2 Sin[2 t], Cos[t] - 2 Cos[2 t]};
tangent = D[trefoil, t];
unitTangent = tangent/Sqrt[tangent.tangent] // Simplify;
normal = D[unitTangent, t];
unitNormal = normal/Sqrt[normal.normal] // Simplify;
biNormal = Cross[unitTangent, unitNormal] // Simplify;
ParametricPlot3D[trefoil + 0.4 Cos[s] unitNormal + 0.4 Sin[s] biNormal,
  {t, 0, 2 Pi}, {s, 0, 2 Pi}, ViewPoint -> {25, 0, 0},
  Boxed -> False, Axes -> False, PlotPoints -> {64, 32}]
```



If you've got *Mathematica* handy, you should try these! Be sure to type the code in *exactly as shown*. Like most computer languages, *Mathematica* is picky about things like case (upper or lower) and punctuation. In particular, all commands are capitalized following a distinctive `TypeOutTheWholeCommand` style. Arguments of all functions and commands are enclosed in square brackets (like this: `[]`). Parentheses are used to group algebraic expressions, as in `x(x^2+1)`. Braces (like this: `{}`) are used to form lists, a fundamental data structure that can represent all sorts of mathematical objects.

## Surface integrals

### Surface area

I wonder how we might find the area of a crazy surface like our lumpy sphere? One approach might be to approximate with parallelograms whose areas are known. This basic idea is shown in figure 8.

We just need some way to compute the areas of those parallelograms in terms of a parametrization of the surface - but we've got it from our study of the cross-product! This is exactly where our main formula comes from:

$$\sum_{i,j=1}^n \|\mathbf{p}_u(u_i, v_j) \times \mathbf{p}_v(u_i, v_j)\| \Delta u \Delta v \rightarrow \iint_D \|\mathbf{p}_u \times \mathbf{p}_v\| \, du dv.$$

Let's illustrate with a couple of computations with *Mathematica*. First, suppose we want to compute the area of a sphere of radius  $r$ . We'd need to integrate the following.

```
p[u_, v_] = r {Cos[u] Sin[v], Sin[u] Sin[v], Cos[v]};
crossNorm = Simplify[Norm[Cross[D[p[u, v], u], D[p[u, v], v]]],
  Assumptions -> {0 < u < 2 Pi, 0 < v < Pi, r > 0}]
r^2 Sin[v]
```

```
Integrate[crossNorm, {u, 0, 2 Pi}, {v, 0, Pi}]
4 π r^2
```

Looks right! The lumpy sphere might be harder.

```
p[u_, v_] = (Sin[5 u] Cos[5 v] + 4) {Cos[u] Sin[v], Sin[u] Sin[v], Cos[v]}/4;
crossNorm = Simplify[Norm[Cross[D[p[u, v], u], D[p[u, v], v]]],
  Assumptions -> {0 < u < 2 Pi, 0 < v < Pi}]
1/64 Sqrt[70 - 45 Cos[2 v] + 6 Cos[8 v] + 13 Cos[10 v] +
  Cos[10 u] (12 + 13 Cos[2 v] - 6 Cos[8 v] + 37 Cos[10 v] - 6 Cos[12 v]) +
  6 Cos[12 v] - 8 Cos[3 v] Sin[5 u] + 16 Cos[5 v] Sin[5 u] - 8 Cos[7 v] Sin[5 u]
  (8 + Sin[5 u - 5 v] + Sin[5 (u + v)])]
```






Figure 8: A crazy surface and some approximations

Hmm... *Mathematica's* `Integrate` command will probably choke on that. It's easy to obtain a numerical estimate, though.

`NIntegrate[crossNorm, {u, -Pi/2, Pi/2}, {v, Pi/12, Pi/2}]`

4.3782

## More general surface integrals

We now ask how to expand these ideas to more general surface integrals. The most general surface integral has the form

$$\int \int_S G(x, y, z) dS = \int_c^d \int_a^b G(x(u, v), y(u, v), z(u, v)) \|\mathbf{r}_u \times \mathbf{r}_v\| du dv.$$

The expression on the left is just notation for the surface integral itself; the expression on the right is a translation of that surface integral to an evaluable iterated integral. In the case of a flux integral, the function  $G$  becomes

$$G(x, y, z) = \mathbf{F}(x, y, z) \cdot \frac{\mathbf{r}_u \times \mathbf{r}_v}{\|\mathbf{r}_u \times \mathbf{r}_v\|}.$$

A flux integral can be written

$$\int \int_S \mathbf{F} \cdot d\mathbf{S} = \int_c^d \int_a^b \mathbf{F}(x(u, v), y(u, v), z(u, v)) \cdot (\mathbf{r}_u \times \mathbf{r}_v) du dv.$$

Let  $T$  denote the top half of the torus, let  $\mathbf{F}(x, y, z) = z$ , and let's use the computer to compute

$$\iint_T \mathbf{F} \cdot d\mathbf{n}.$$

Here's the parametrization:

$$\begin{aligned} \mathbf{p}[\mathbf{s}_-, \mathbf{t}_-] &= \{\text{Cos}[\mathbf{t}], \text{Sin}[\mathbf{t}], 0\} + \{\text{Cos}[\mathbf{t}], \text{Sin}[\mathbf{t}], 0\} \text{Cos}[\mathbf{s}] / 2 + \{0, 0, 1\} \text{Sin}[\mathbf{s}] / 2 \\ &\left\{ \text{Cos}[\mathbf{t}] + \frac{1}{2} \text{Cos}[\mathbf{s}] \text{Cos}[\mathbf{t}], \text{Sin}[\mathbf{t}] + \frac{1}{2} \text{Cos}[\mathbf{s}] \text{Sin}[\mathbf{t}], \frac{\text{Sin}[\mathbf{s}]}{2} \right\} \end{aligned}$$

Since  $\mathbf{F}(x, y, z) = \langle 0, 0, z \rangle$ , we know that  $\mathbf{F} = \langle 0, 0, \sin(s)/2 \rangle$  on the surface. Here's the (un-normalized) normal vector.

`normal = Simplify[Cross[D[p[s, t], t], D[p[s, t], s]]]`

$$\begin{aligned} &\left\{ \frac{1}{8} (1 + 4 \text{Cos}[\mathbf{s}] + \text{Cos}[2 \mathbf{s}]) \text{Cos}[\mathbf{t}], \right. \\ &\left. \frac{1}{8} (1 + 4 \text{Cos}[\mathbf{s}] + \text{Cos}[2 \mathbf{s}]) \text{Sin}[\mathbf{t}], \frac{1}{8} (4 \text{Sin}[\mathbf{s}] + \text{Sin}[2 \mathbf{s}]) \right\} \end{aligned}$$

Here's the integrand.

$$\mathbf{integrand} = \{0, 0, \text{Sin}[s] / 2\} \cdot \mathbf{normal}$$

$$\frac{1}{16} \text{Sin}[s] (4 \text{Sin}[s] + \text{Sin}[2s])$$

$$\mathbf{Integrate}[\mathbf{integrand}, \{s, 0, \text{Pi}\}, \{t, 0, 2 \text{Pi}\}]$$

$$\frac{\pi^2}{4}$$

## Exercises

1. Use a 3D parametric plotter to generate the top half a sphere.
2. The tropics of Cancer and Capricorn have latitude approximately  $23.5^\circ \approx 0.13\pi$ . Find a parametrization of the portion of the sphere of radius 1 between these tropics and use a computer to plot the result.
3. Find a parametrization of the portion of the graph of  $z = \sin(x^2 + y^2)$  over the top half of the unit disk and use a computer to plot the result.
4. Try all of the examples in the “Generating parametric plots on the computer” section.
5. In this problem, we work with the parametric function  $\mathbf{p}(u, v) = (\cos(u), \sin(u), v)$ .
  - (a) Use a 3D parametric plotter to plot the image of the rectangle  $[0, 2\pi] \times [0, 5]$  in the  $uv$ -plane after application of  $\mathbf{p}$ . You should recognize the figure and it should make some good sense.
  - (b) Compute  $\|\mathbf{p}_u \times \mathbf{p}_v\|$ .
  - (c) Use part (b) to compute the area of the surface you graphed in part (a)
6. Using the same surface as in problem 5, compute

$$\iint_T \mathbf{F} \cdot d\mathbf{n},$$

where  $\mathbf{F}(x, y, z) = \langle x, y, 0 \rangle$ .